Usean yhtälön ryhmän ratkaiseminen yhteen yhtälöön palauttamalla

Olkoon tarkasteltavana kolmen differentiaaliyhtälön normaalimuotoinen ryhmä, jossa tuntemattomina funktioina ovat x(*t*), y(*t*) ja z(*t*):

```
> dy1:= diff(x(t), t)=x(t)+y(t)-z(t)+1;
  dy2:= diff(y(t), t)=-x(t)+5*y(t)+z(t)+t;
  dy3:= diff(z(t), t)=-2*x(t)+2*y(t)+4*z(t)+t^2;
```

$$dy1 := \frac{d}{dt}\,\mathrm{x}(t) = \mathrm{x}(t) + \mathrm{y}(t) - \mathrm{z}(t) + 1$$

$$dy2 := \frac{d}{dt}\,\mathrm{y}(t) = -\mathrm{x}(t) + 5\,\mathrm{y}(t) + \mathrm{z}(t) + t$$

$$dy3 := \frac{d}{dt}\,\mathrm{z}(t) = -2\,\mathrm{x}(t) + 2\,\mathrm{y}(t) + 4\,\mathrm{z}(t) + t^2$$

```
> funktiot:= [x(t), y(t), z(t)];
```

$$funktiot := [\mathrm{x}(t), \mathrm{y}(t), \mathrm{z}(t)]$$

Ryhmä pyritään ratkaisemaan eliminoimalla ensin yhtälöistä funktiot *x* ja *y*, jolloin saadaan yksinomaan funktiota *z* koskeva differentiaaliyhtälö. Tätä varten kaksi ensimmäistä yhtälöä derivoidaan kerran ja viimeinen yhtälö kaksi kertaa, jolloin saadaan kaikkiaan seitsemän yhtälöä:

```
> dy4:= diff(dy1, t):
  dy5:= diff(dy2, t):
  dy6:= diff(dy3, t):
  dy7:= diff(dy3, t$2):
  dy1, dy2, dy3, dy4, dy5, dy6, dy7;
```

$$\frac{d}{dt}\,\mathrm{x}(t) = \mathrm{x}(t) + \mathrm{y}(t) - \mathrm{z}(t) + 1,\ \frac{d}{dt}\,\mathrm{y}(t) = -\mathrm{x}(t) + 5\,\mathrm{y}(t) + \mathrm{z}(t) + t,$$

$$\frac{d}{dt}\,\mathrm{z}(t) = -2\,\mathrm{x}(t) + 2\,\mathrm{y}(t) + 4\,\mathrm{z}(t) + t^2,\ \frac{d^2}{dt^2}\,\mathrm{x}(t) = \left(\frac{d}{dt}\,\mathrm{x}(t)\right) + \left(\frac{d}{dt}\,\mathrm{y}(t)\right) - \left(\frac{d}{dt}\,\mathrm{z}(t)\right),$$

$$\frac{d^2}{dt^2}\,\mathrm{y}(t) = -\left(\frac{d}{dt}\,\mathrm{x}(t)\right) + 5\left(\frac{d}{dt}\,\mathrm{y}(t)\right) + \left(\frac{d}{dt}\,\mathrm{z}(t)\right) + 1,$$

$$\frac{d^2}{dt^2}\,\mathrm{z}(t) = -2\left(\frac{d}{dt}\,\mathrm{x}(t)\right) + 2\left(\frac{d}{dt}\,\mathrm{y}(t)\right) + 4\left(\frac{d}{dt}\,\mathrm{z}(t)\right) + 2\,t,$$

$$\frac{d^3}{dt^3}\,\mathrm{z}(t) = -2\left(\frac{d^2}{dt^2}\,\mathrm{x}(t)\right) + 2\left(\frac{d^2}{dt^2}\,\mathrm{y}(t)\right) + 4\left(\frac{d^2}{dt^2}\,\mathrm{z}(t)\right) + 2$$

Kuudesta ensimmäisestä yhtälöstä voidaan ratkaista funktiot *x* ja *y* derivaattoineen ja tämän jälkeen sijoittaa lausekkeet perättäin viimeiseen yhtälöön.

Määritellään ensin tuntemattomat:

```
> tuntemattomat:=[diff(y(t), t, t), diff(x(t), t, t), diff(x(t), t), diff(y(t), t), y(t), x(t)];
```

$$tuntemattomat := \left[\frac{d^2}{dt^2}\,\mathrm{y}(t), \frac{d^2}{dt^2}\,\mathrm{x}(t), \frac{d}{dt}\,\mathrm{x}(t), \frac{d}{dt}\,\mathrm{y}(t), \mathrm{y}(t), \mathrm{x}(t)\right]$$

Ratkaistaessa *Maple*n **solve**-komennolla ei ole mahdollista ratkaista derivaattojen suhteen, jos derivaatasta esiintyy useampia kertalukuja. Tämä voidaan ohittaa korvaamalla derivaatta- ja funktiomerkinnät väliaikaisilla muuttujilla.

```
> d2m:= zip((x, y)->x=y, tuntemattomat, [seq(d[k], k=1..6)]):
  m2d:= zip((x, y)->x=y, [seq(d[k], k=1..6)], tuntemattomat):
> subs(d2m, [{dy1, dy2, dy3, dy4, dy5, dy6}, tuntemattomat]):
  solve(%[1], {%[2][]}):
  elimsij:= subs(m2d, %);
```

$$elimsij := \Big\{\frac{d^2}{dt^2}\,\mathrm{x}(t) = \frac{3}{2}\left(\frac{d^2}{dt^2}\,\mathrm{z}(t)\right) - 10\left(\frac{d}{dt}\,\mathrm{z}(t)\right) + 6\,\mathrm{z}(t) + 3\,t^2 + 4 - 5\,t,$$

$$\frac{d^2}{dt^2}\,\mathrm{y}(t) = \frac{9}{2}\left(\frac{d^2}{dt^2}\,\mathrm{z}(t)\right) - 23\left(\frac{d}{dt}\,\mathrm{z}(t)\right) + 12\,\mathrm{z}(t) + 6\,t^2 + 9 - 13\,t,$$

$$\frac{d}{dt}\,\mathrm{y}(t) = \left(\frac{d^2}{dt^2}\,\mathrm{z}(t)\right) - \frac{11}{2}\left(\frac{d}{dt}\,\mathrm{z}(t)\right) + 3\,\mathrm{z}(t) + \frac{3\,t^2}{2} + 2 - 3\,t,$$

$$\mathrm{y}(t) = -\frac{3}{2}\left(\frac{d}{dt}\,\mathrm{z}(t)\right) + \frac{1}{4}\left(\frac{d^2}{dt^2}\,\mathrm{z}(t)\right) + \mathrm{z}(t) + \frac{t^2}{2} + \frac{1}{2} - t,$$

$$\frac{d}{dt}\,\mathrm{x}(t) = \frac{1}{2}\left(\frac{d^2}{dt^2}\,\mathrm{z}(t)\right) - \frac{7}{2}\left(\frac{d}{dt}\,\mathrm{z}(t)\right) + 3\,\mathrm{z}(t) + \frac{3\,t^2}{2} + 2 - 2\,t,$$

$$\mathrm{x}(t) = \frac{1}{4}\left(\frac{d^2}{dt^2}\,\mathrm{z}(t)\right) - 2\left(\frac{d}{dt}\,\mathrm{z}(t)\right) + 3\,\mathrm{z}(t) + t^2 + \frac{1}{2} - t\Big\}$$

```
> dy:= subs(elimsij, dy7);
```

$$dy := \frac{d^3}{dt^3}\,\mathrm{z}(t) = 10\left(\frac{d^2}{dt^2}\,\mathrm{z}(t)\right) - 26\left(\frac{d}{dt}\,\mathrm{z}(t)\right) + 12\,\mathrm{z}(t) + 6\,t^2 + 12 - 16\,t$$

Tuloksena on kolmannen kertaluvun yhtälö funktiolle *z*. Tämä voidaan löytää myös hieman suorempaan käyttämällä *Maple*n **eliminate**-komentoa. Myös **eliminate**-komennon käyttö vaatii edellä esitellyt korvaustoimenpiteet.

```
> subs(d2m, [{dy1, dy2, dy3, dy4, dy5, dy6, dy7},
  tuntemattomat]):
  eliminate(%[1], {%[2][]}):
  op(subs(m2d, %)[-1])=0;
```

$$-\left(\frac{d^3}{dt^3}\mathrm{z}(t)\right)+10\left(\frac{d^2}{dt^2}\mathrm{z}(t)\right)-26\left(\frac{d}{dt}\mathrm{z}(t)\right)+12\,\mathrm{z}(t)+6\,t^2+12-16\,t=0$$

Yhtälön ratkaisuksi saadaan

```
> zratk:= dsolve(dy, z(t));
```

$$zratk := \mathrm{z}(t) = -\frac{71}{36}-\frac{5\,t}{6}-\frac{t^2}{2}+\_C1\,\mathbf{e}^{(6\,t)}+\_C2\,\mathbf{e}^{((2+\sqrt{2})\,t)}+\_C3\,\mathbf{e}^{(-(-2+\sqrt{2})\,t)}$$

Yhtälöryhmän yleinen ratkaisu muodossa x(t), y(t), z(t) saadaan tätä ja aiempia lausekkeita käyttäen:

```
> subs(elimsij, funktiot):
  ylratk:= subs(zratk, %):
  simplify(%);
```

$$\left[-4-\_C2\,\mathbf{e}^{((2+\sqrt{2})\,t)}\sqrt{2}+\frac{1}{2}\_C2\,\mathbf{e}^{((2+\sqrt{2})\,t)}+\_C3\,\mathbf{e}^{(-(-2+\sqrt{2})\,t)}\sqrt{2}+\frac{1}{2}\_C3\,\mathbf{e}^{(-(-2+\sqrt{2})\,t)}-\frac{3\,t}{2}-\frac{t^2}{2},\ -\frac{17}{36}-\frac{t}{3}+\_C1\,\mathbf{e}^{(6\,t)}-\frac{1}{2}\_C2\,\mathbf{e}^{((2+\sqrt{2})\,t)}-\frac{1}{2}\_C2\,\mathbf{e}^{((2+\sqrt{2})\,t)}\sqrt{2}-\frac{1}{2}\_C3\,\mathbf{e}^{(-(-2+\sqrt{2})\,t)}+\frac{1}{2}\_C3\,\mathbf{e}^{(-(-2+\sqrt{2})\,t)}\sqrt{2},\ -\frac{71}{36}-\frac{5\,t}{6}-\frac{t^2}{2}+\_C1\,\mathbf{e}^{(6\,t)}+\_C2\,\mathbf{e}^{((2+\sqrt{2})\,t)}+\_C3\,\mathbf{e}^{(-(-2+\sqrt{2})\,t)}\right]$$

Tiettyä alkuehtoa vastaavat vakioiden arvot saadaan tämän jälkeen tavalliseen tapaan ratkaisemalla algebrallinen yhtälöryhmä. Olkoon alkuehtona x(0) = 1, y(0) = -2, z(0) = 3. Tällöin

```
> subs(t=0, ylratk):
  vakiot:= solve({%[1]=1, %[2]=-2, %[3]=3}, {_C1, _C2, _C3}):
  simplify(%);
```

$$\left\{\_C1=\frac{-43}{252},\ \_C3=\frac{18}{7}+\frac{17\sqrt{2}}{28},\ \_C2=-\frac{17\sqrt{2}}{28}+\frac{18}{7}\right\}$$

Vastaava yksittäisratkaisu saadaan sijoittamalla vakioiden arvot yleiseen ratkaisuun:

```
> yksittratk:= subs(vakiot, ylratk):
  simplify(%);
```

$$\left[-4-\frac{23}{8}\mathbf{e}^{((2+\sqrt{2})\,t)}\sqrt{2}+\frac{5}{2}\mathbf{e}^{((2+\sqrt{2})\,t)}+\frac{23}{8}\mathbf{e}^{(-(-2+\sqrt{2})\,t)}\sqrt{2}+\frac{5}{2}\mathbf{e}^{(-(-2+\sqrt{2})\,t)}-\frac{3\,t}{2}-\frac{t^2}{2},\ -\frac{17}{36}-\frac{t}{3}-\frac{43}{252}\mathbf{e}^{(6\,t)}-\frac{55}{56}\mathbf{e}^{((2+\sqrt{2})\,t)}\sqrt{2}-\frac{19}{28}\mathbf{e}^{((2+\sqrt{2})\,t)}-\frac{19}{28}\mathbf{e}^{(-(-2+\sqrt{2})\,t)}+\frac{55}{56}\mathbf{e}^{(-(-2+\sqrt{2})\,t)}\sqrt{2},\ -\frac{71}{36}-\frac{5\,t}{6}-\frac{t^2}{2}-\frac{43}{252}\mathbf{e}^{(6\,t)}-\frac{17}{28}\mathbf{e}^{((2+\sqrt{2})\,t)}\sqrt{2}+\frac{18}{7}\mathbf{e}^{((2+\sqrt{2})\,t)}+\frac{18}{7}\mathbf{e}^{(-(-2+\sqrt{2})\,t)}\right.$$

$$+\frac{17}{28}\,\mathbf{e}^{(-(-2+\sqrt{2})\,t)}\sqrt{2}\,\Big]$$

Ratkaisujen kuvaajat:

```
> plot(yksittratk, t=0..1, view=[0..1, -20..20], color=red);
```



Edellä oleva esitys kuvaa, miten eliminointiprosessi ja yhtälöryhmän ratkaiseminen tapahtuu. Jos tavoitteena on ainoastaan saada tietyn alkuarvoprobleeman ratkaisu, päästään paljon vähemmällä kohdistamalla *Maple*n **dsolve**-funktio suoraan alkuperäiseen yhtälöryhmään ja alkuehtoon:

```
> suoraratk:= dsolve({dy1, dy2, dy3, x(0)=1, y(0)=-2,
  z(0)=3},funktiot):
  simplify(%);
```

$$\Big\{\,z(t)=-\frac{71}{36}-\frac{5\,t}{6}-\frac{t^2}{2}-\frac{43}{252}\,\mathbf{e}^{(6\,t)}-\frac{17}{28}\,\mathbf{e}^{((2+\sqrt{2})\,t)}\sqrt{2}+\frac{18}{7}\,\mathbf{e}^{((2+\sqrt{2})\,t)}+\frac{18}{7}\,\mathbf{e}^{(-(-2+\sqrt{2})\,t)}$$
$$+\frac{17}{28}\,\mathbf{e}^{(-(-2+\sqrt{2})\,t)}\sqrt{2},\ y(t)=$$
$$-\frac{17}{36}-\frac{t}{3}-\frac{43}{252}\,\mathbf{e}^{(6\,t)}-\frac{55}{56}\,\mathbf{e}^{((2+\sqrt{2})\,t)}\sqrt{2}-\frac{19}{28}\,\mathbf{e}^{((2+\sqrt{2})\,t)}-\frac{19}{28}\,\mathbf{e}^{(-(-2+\sqrt{2})\,t)}+\frac{55}{56}\,\mathbf{e}^{(-(-2+\sqrt{2})\,t)}\sqrt{2}$$
$$,\ x(t)=-4-\frac{23}{8}\,\mathbf{e}^{((2+\sqrt{2})\,t)}\sqrt{2}+\frac{5}{2}\,\mathbf{e}^{((2+\sqrt{2})\,t)}+\frac{23}{8}\,\mathbf{e}^{(-(-2+\sqrt{2})\,t)}\sqrt{2}+\frac{5}{2}\,\mathbf{e}^{(-(-2+\sqrt{2})\,t)}-\frac{3\,t}{2}-\frac{t^2}{2}\Big\}$$

Eri tavoilla saadut ratkaisut ovat todellakin samat:

```
> subs(suoraratk, [x(t), y(t), z(t)]):
  zip((x, y)->simplify(x-y), %, yksittratk);
```

$$[0, 0, 0]$$

```
>
```

Linkit

differentiaaliyhtälöryhmä
ryhmän palauttaminen yhteen yhtälöön
differentiaaliyhtälön ratkaiseminen symbolisella ohjelmalla (symalg.mws)

SKK & MS 03.01.2001